

This copy of the Lingua Franca handbook for the rs target was created on Sunday, September 3, 2023 against commit <a href="mailto:a5b0ce">a5b0ce</a>.

#### **Table of Contents**

<u>A First Reactor</u> Writing your first Lingua Franca reactor.

<u>Inputs and Outputs</u> Inputs, outputs, and reactions in Lingua Franca.

Parameters and State

Parameters and state variables in Lingua Franca.

<u>Variables</u>

<u>Time and Timers</u>

<u>Composing Reactors</u>

<u>Reactions and Methods</u>

Time and timers in Lingua Franca.

Composing reactors in Lingua Franca.

Reactions and methods in Lingua Franca.

<u>Causality Loops</u> Causality loops in Lingua Franca. <u>Extending Reactors</u> Extending reactors in Lingua Franca.

<u>Actions</u> Actions in Lingua Franca.

<u>Superdense Time</u> Superdense time in Lingua Franca.

Modal Reactors Modal Reactors

<u>Deadlines</u> Deadlines in Lingua Franca.

<u>Multiports and Banks</u> Multiports and Banks of Reactors.

Generic Reactors Defining generic reactors in Lingua Franca.

<u>Preambles</u> Defining preambles in Lingua Franca.

<u>Distributed Execution</u> Distributed Execution (preliminary)

<u>Termination</u> Terminating a Lingua Franca execution.

#### **A First Reactor**

See the <u>requirements</u> for using this target.

## Minimal Example

A minimal but complete Lingua Franca file with one reactor is this:

```
target Rust
main reactor {
  reaction(startup) {=
    println!("Hello World.");
  =}
}
```

Every Lingua Franca program begins with a <u>target declaration</u> that specifies the language in which reactions are written. This is also the language of the program(s) generated by the Lingua Franca code generator.

Every LF program also has a main reactor, which is the top level of a hierarchy of contained and interconnected reactors. The above simple example has no contained reactors.

The main reactor above has a single reaction, which is triggered by the startup trigger. This trigger causes the reaction to execute at the start of the program. The body of the reaction, delimited by  $\{= \ldots =\}$ , is ordinary Rust code which, as we will see, has access to a number of functions and variables specific to Lingua Franca.

## Examples

Examples of Lingua Franca programs can be found in the Lingua Franca Playground.

The <u>regression tests</u> have a rich set of examples that illustrate every feature of the language.

# Structure of an LF Project

The Lingua Franca tools assume that LF programs are put into a file with a .lf extension that is stored somewhere within a directory called src. To compile and run the above example, choose a **project root** directory, create a src directory within that, and put the above code into a file

called, say, <a href="scriptor">src/HelloWorld.lf</a>. You can compile the code on the <a href="command line">command line</a>, within <a href="yisual">yisual</a> <a href="yisual">Studio Code</a>, or within the <a href="Epoch IDE">Epoch IDE</a>. On the command line this will look like this:

```
> lfc src/HelloWorld.lf
... output from the code generator and compiler ...
```

After this completes, two additional directories will have been created within the project root, bin and src-gen. The bin directory has an executable file called HelloWorld. Executing that file will result, not surprisingly, in printing "Hello World". The generated source files will be in a subdirectory called HelloWorld within src-gen.

#### Reactor Block

A **reactor** is a software component that reacts to input events, timer events, and internal events. It has private state variables that are not visible to any other reactor. Its reactions can consist of altering its own state, sending messages to other reactors, or affecting the environment through some kind of actuation or side effect (e.g., printing a message, as in the above HelloWorld example).

The general structure of a reactor definition is as follows:

```
[main] reactor <class-name> [(<parameters>)] {
    input <name>:<type>
    output <name>:<type>
    state <name>:<type>(<value>)
    timer <name>([<offset>, [<period>]])
    logical action <name>[:<type>]
    physical action <name>[:<type>]
    [const] method <name>(<parameters>):<type> {= ... body ...=}
    reaction(<triggers>) [<uses>] [=> <effects>] {= ... body ...=}
    <instance-name> = new <class-name>([<parameter-assignments>])
    <instance-name> [, ...] => <instance-name> [, ...] [after <delay>]
}
```

Contents within square brackets are optional, contents within <...> are user-defined, and each line may appear zero or more times, as explained in the next pages. Parameters, inputs, outputs, timers, actions, and contained reactors all have names, and the names are required to be distinct from one another.

If the **reactor** keyword is preceded by **main**, then this reactor will be instantiated and run by the generated code.

Any number of reactors may be defined in one file, and a main reactor need not be given a name, but if it is given a name, then that name must match the file name.

Reactors may extend other reactors, inheriting their properties, and a file may import reactors from other files. If an imported LF file contains a **main** reactor, that reactor is ignored (it will not be imported). This makes it easy to create a library of reusable reactors that each come with a test case or demonstration in the form of a main reactor.

#### Comments

Lingua Franca files can have C/C++/Java-style comments and/or Python-style comments. All of the following are valid comments:

```
// Single-line C-style comment.
/*
 * Multi-line C-style comment.
 */
# Single-line Python-style comment.
```

# **Inputs and Outputs**

In this section, we will endow reactors with inputs and outputs.

# Input and Output Declarations

Input and output declarations have the form:

```
input <name>:<type>
output <name>:<type>
```

For example, the following reactor doubles its input and sends the result to the output:

```
target Rust
reactor Double {
  input x: u32
  output y: u32
  reaction(x) -> y {=
    ctx.set(y, ctx.get(x).unwrap() * 2);
  =}
}
```

Notice how the input value is accessed and how the output value is set. This is done differently for each target language. See the <u>Target Language Details</u> for detailed documentation of these mechanisms. Setting an output within a reaction will trigger downstream reactions at the same <u>Logical Time</u> that the reaction is invoked (or, more precisely, at the same <u>tag</u>). If a particular output port is set more than once at any tag, the last set value will be the one that downstream reactions see. Since the order in which reactions of a reactor are invoked at a logical time is deterministic, and whether inputs are present depends only on their timestamps, the final value set for an output will also be deterministic.

The **type** of a port is a type in the target language plus the special type **time**. A type may also be specified using a **code block**, delimited by the same delimiters  $\{= \ldots =\}$  that separate target language code from Lingua Franca code in reactions. Any valid target-language type designator can be given within these delimiters.

The **reaction** declaration above indicates that an input event on port x is a **trigger** and that an output event on port y is a (potential) **effect**. A reaction can declare more than one trigger or effect by just listing them separated by commas. For example, the following reactor has two triggers and tests each input for presence before using it:

```
target Rust
reactor Destination {
  input x: u32
  input y: u32
  reaction(x, y) {=
    let mut sum = 0;
    if let Some(x) = ctx.get(x) {
       sum += x;
    }
    if let Some(y) = ctx.get(y) {
       sum += y;
    }
    println!("Received {}.", sum);
    =}
}
```

**NOTE:** if a reaction fails to test for the presence of an input and reads its value anyway, then the result it will get is target dependent. FIXME.

# Triggers, Effects, and Uses

The general form of a reaction is

```
reaction (<triggers>) <uses> -> <effects> {=
    <target language code>
    =}
```

The **triggers** field can be a comma-separated list of input ports, <u>output ports of contained</u> <u>reactors</u>, <u>timers</u>, <u>actions</u>, or the special events <u>startup</u>, <u>shutdown</u>, and <u>reset</u>. There must be at least one trigger for each reaction. For the special events, see the <u>Reactions and Methods</u> section.

The **uses** field, which is optional, specifies input ports (or output ports of contained reactors) that do not trigger execution of the reaction but may be read by the reaction.

The **effects** field, which is also optional, is a comma-separated lists of output ports ports, input ports of contained reactors, or <u>actions</u>.

# Setting an Output Multiple Times

If one or more reactions set an output multiple times at the same  $\underline{tag}$ , then only the last value set will be seen by any downstream reactors.

If a reaction wishes to test whether an output has been previously set at the current tag by some other reaction, it can test it in the same way it tests inputs for presence.

## Mutable Inputs

Normally, a reaction does not modify the value of an input. An input is said to be **immutable**. The degree to which this is enforced varies by target language. Most of the target languages make it rather difficult to enforce, so the programmer needs to avoid modifying the input. Modifying an input value may lead to nondeterministic results.

Occasionally, it is useful to modify an input. For example, the input may be a large data structure, and a reaction may wish to make a small modification and forward the result to an output. To accomplish this, the programmer should declare the input **mutable** as follows:

```
mutable input <name>:<type>
```

This is a directive to the code generator indicating that reactions that read this input may also modify the value of the input. The code generator will attempt to optimize the scheduling to avoid copying the input value, but this may not be possible, in which case it will automatically insert a copy operation, making it safe to modify the input. The target-specific reference documentation has more details about how this works.

#### **Parameters and State Variables**

#### Parameter Declaration

A reactor class definition can parameterized as follows:

```
reactor <class-name>(<param-name>:<type> = <expr>, ...) {
    ...
}
```

Each parameter has a *type annotation*, written :<type> , where <type> has one of the following forms:

- An identifier, such as int.
- An array type type[] and type[integer].
- The keyword time, which designates a time value.
- A code block delimited by {= ... =}, where the contents is any valid type in the target language.

Depending on the target, the type may be a <u>generic type</u>, which means that the type is parameter determined at the time the reactor class is instantiated.

For example, the Double reactor on the <u>previous page</u> can be replaced with a more general parameterized reactor Scale as follows:

```
target Rust
reactor Scale(factor: u32 = 2) {
   state factor = factor
   input x: u32
   output y: u32
   reaction(x) -> y {=
     let x = ctx.get(x).unwrap();
     ctx.set(y, x * self.factor);
   =}
}
```

This reactor, given any input event x will produce an output y with value equal to the input scaled by the factor parameter. The default value of the factor parameter is 2, but this can be changed when the Scale reactor is instantiated.

Notice how, within the body of a reaction, the code accesses the parameter value. This is different for each target language.

#### State Declaration

A reactor declares a state variable as follows:

```
state <name>:<type> = <value>
```

The type can any of the same forms as for a parameter.

The <value> is an initial value and, like parameter values, can be given as an <u>expression</u> or target language code with delimiters {= ... =}. The initial value can also be given as a parameter name. The value can be accessed and modified in a target-language-dependent way as illustrated by the following example:

```
target Rust
reactor Count {
   state count: u32 = 0
   output y: u32
   timer t(0, 100 msec)
   reaction(t) -> y {=
     ctx.set(y, self.count);
     self.count += 1;
   =}
}
```

This reactor has an integer state variable named **count**, and each time its reaction is invoked, it outputs the value of that state variable and increments it. The reaction is triggered by a **timer**, discussed in the next section.

#### Reset State Variables

The reset keyword is not supported in Rust because modal reactors are not supported.

#### **Time and Timers**

# **Logical Time**

A key property of Lingua Franca is **logical time**. All events occur at an instant in logical time. By default, the runtime system does its best to align logical time with **physical time**, which is some measurement of time on the execution platform. The **lag** is defined to be physical time minus logical time, and the goal of the runtime system is maintain a small non-negative lag.

The **lag** is allowed to go negative only if the <u>fast target property</u> or the <u>--fast</u> command-line argument is set to true. In that case, the program will execute as fast as possible with no regard to physical time.

In Lingua Franca, time is a data type. A parameter, state variable, port, or action may have type time. In the Rust target, time values internally have type FIXME.

#### Time Values

A time value is given with units (unless the value is 0, in which case the units can be omitted). The allowable units are:

- For nanoseconds: ns , nsec , or nsecs
- For microseconds: us , usec , or usecs
- For milliseconds: ms , msec , or msecs
- For seconds: s, sec, secs, second, or seconds
- For minutes: min, minute, mins, or minutes
- For hours: h, hour, or hours
- For days: d, day, or days
- For weeks: week or weeks

The following example illustrates using time values for parameters and state variables:

```
target Rust
main reactor SlowingClock(start: time = 100 ms, incr: time = 100 ms) {
  state start = start
  state incr = incr
  state interval: time = start
  state expected time: time()
  logical action a
  reaction(startup) -> a {=
    ctx.schedule(a, After(self.start));
 =}
  reaction(a) -> a {=
    println!(
        "Logical time since start: {} nsec.",
        ctx.get elapsed logical time().as nanos(),
    );
    self.interval += self.incr;
    ctx.schedule(a, After(self.interval));
    self.expected time += self.interval;
  =}
}
```

This has two time parameters, start and incr, each with default value 100 ms and type time. This parameter is used to initialize the interval state variable, which also stores a time. The logical action a, explained in Actions, is used to schedule events to occur at time start after program startup and then at intervals that are increased each time by incr. The result of executing this program will look like this:

```
Logical time since start: 100000000 nsec.
Logical time since start: 300000000 nsec.
Logical time since start: 600000000 nsec.
Logical time since start: 1000000000 nsec.
...
```

#### **Timers**

The simplest use of logical time in Lingua Franca is to invoke a reaction periodically. This is done by first declaring a timer using this syntax:

```
timer <name>(<offset>, <period>)
```

The <period>, which is optional, specifies the time interval between timer events. The <offset>, which is also optional, specifies the (logical) time interval between when the program starts executing and the first timer event. If no period is given, then the timer event occurs only

once. If neither an offset nor a period is specified, then one timer event occurs at program start, simultaneous with the **startup** event.

The period and offset are given by a number and a units, for example, 10 ms. See the expressions documentation for allowable units. Consider the following example:

```
target Rust
main reactor Timer {
  timer t(0, 1 sec)
  reaction(t) {=
    println!(
       "Logical time is {}.",
       ctx.get_elapsed_logical_time().as_nanos(),
    );
  =}
}
```

This specifies a timer named t that will first trigger at the start of execution and then repeatedly trigger at intervals of one second. Notice that the time units can be left off if the value is zero.

This target provides a built-in function for retrieving the logical time at which the reaction is invoked, FIXME. On most platforms (with the exception of some embedded platforms), the returned value is a 64-bit number representing the number of nanoseconds that have elapsed since January 1, 1970. Executing the above displays something like the following:

```
Logical time is 1648402121312985000.
Logical time is 1648402122312985000.
Logical time is 1648402123312985000.
```

The output lines appear at one second intervals unless the fast option has been specified.

## **Elapsed Time**

The times above are a bit hard to read, so, for convenience, each target provides a built-in function to retrieve the *elapsed* time. For example:

```
target Rust
main reactor TimeElapsed {
  timer t(0, 1 s)
  reaction(t) {=
    println!(
        "Elapsed logical time is {}.",
        ctx.get_elapsed_logical_time().as_nanos(),
    );
    =}
}
```

See the <u>Target Language Details</u> for the full set of functions provided for accessing time values.

Executing this program will produce something like this:

```
Elapsed logical time is 0.
Elapsed logical time is 1000000000.
Elapsed logical time is 2000000000.
```

# Comparing Logical and Physical Times

The following program compares logical and physical times:

```
target Rust
main reactor TimeLag {
  timer t(0, 1 s)
  reaction(t) {=
    let t = ctx.get_elapsed_logical_time();
    let T = ctx.get_elapsed_physical_time();
    println!(
        "Elapsed logical time: {}, physical time: {}, lag: {}",
        t.as_nanos(),
        T.as_nanos(),
        (T-t).as_nanos(),
    );
    =}
}
```

Execution will show something like this:

```
Elapsed logical time: 0, physical time: 855000, lag: 855000
Elapsed logical time: 1000000000, physical time: 1004714000, lag: 4714000
Elapsed logical time: 2000000000, physical time: 2004663000, lag: 4663000
Elapsed logical time: 3000000000, physical time: 3000210000, lag: 210000
```

In this case, the lag varies from a few hundred microseconds to a small number of milliseconds. The amount of lag will depend on the execution platform.

## Simultaneity and Instantaneity

If two timers have the same *offset* and *period*, then their events are logically simultaneous. No observer will be able to see that one timer has triggered and the other has not.

A reaction is always invoked at a well-defined logical time, and logical time does not advance during its execution. Any output produced by the reaction will be **logically simultaneous** with the input. In other words, reactions are **logically instantaneous** (for an exception, see <u>Logical Execution Time</u>). Physical time, however, does elapse during execution of a reaction.

# **Timeout**

By default, a Lingua Franca program will terminate when there are no more events to process. If there is a timer with a non-zero period, then there will always be more events to process, so the default execution will be unbounded. To specify a finite execution horizon, you can either specify a <a href="timeout target property">timeout target property</a> or a <a href="timeout command-line option">- timeout command-line option</a>. For example, the following timeout property will cause the above timer with a period of one second to terminate after 11 events:

```
target Rust {
  timeout: 10 s
}
```

## Startup and Shutdown

To cause a reaction to be invoked at the start of execution, a special **startup** trigger is provided:

```
reactor Foo {
  reaction(startup) {=
    ... perform initialization ...
  =}
}
```

The **startup** trigger is equivalent to a timer with no *offset* or *period*.

To cause a reaction to be invoked at the end of execution, a special **shutdown** trigger is provided. Consider the following reactor, commonly used to build regression tests:

```
target Rust
reactor TestCount(start: u32 = 0, stride: u32 = 1, num inputs: u32 = 1) {
  state stride = stride
  state num inputs = num_inputs
  state count: u32 = start
  state inputs received: u32 = 0
  input x: u32
  reaction(x) {=
    let x = ctx.get(x).unwrap();
    println!("Received {}.", x);
    if x != self.count {
      println!("ERROR: Expected {}.", self.count);
      std::process::exit(1);
    self.count += self.stride;
    self.inputs received += 1;
 =}
  reaction(shutdown) {=
    println!("Shutdown invoked.");
    if self.inputs received != self.num inputs {
      println!(
          "ERROR: Expected to receive {} inputs, but got {}.",
          self.num inputs,
          self.inputs received
      );
      std::process::exit(2);
   }
  =}
}
```

This reactor tests its inputs against expected values, which are expected to start with the value given by the start parameter and increase by stride with each successive input. It expects to receive a total of num\_inputs input events. It checks the total number of inputs received in its shutdown reaction.

The **shutdown** trigger typically occurs at <u>microstep</u> 0, but may occur at a larger microstep. See <u>Superdense Time</u> and <u>Termination</u>.

# **Composing Reactors**

#### **Contained Reactors**

Reactors can contain instances of other reactors defined in the same file or in an imported file.

Assume the Count and Scale reactors defined in <u>Parameters and State Variables</u> are stored in files Count.lf and Scale.lf, respectively, and that the <u>TestCount</u> reactor from <u>Time and</u> <u>Timers</u> is stored in <u>TestCount.lf</u>. Then the following program composes one instance of each of the three:

```
target Rust {
  timeout: 1 sec,
  fast: true
}
import Count from "Count.lf"
import Scale from "Scale.lf"
import TestCount from "TestCount.lf"
main reactor RegressionTest {
  c = new Count()
  s = new Scale(factor=4)
  t = new TestCount(stride=4, num_inputs=11)
  c.y -> s.x
  s.y -> t.x
}
```

#### **Diagrams**

As soon as programs consist of more than one reactor, it becomes particularly useful to reference the diagrams that are automatically created and displayed by the Lingua Franca IDEs. The diagram for the above program is as follows:



In this diagram, the timer is represented by a clock-like icon, the reactions by chevron shapes, and the **shutdown** event by a diamond. If there were a **startup** event in this program, it would appear as a circle.

#### **Creating Reactor Instances**

An instance is created with the syntax:

```
<instance_name> = new <class_name>(<parameters>)
```

A bank with several instances can be created in one such statement, as explained in the <u>banks</u> of reactors documentation.

The <parameters> argument is a comma-separated list of assignments:

```
<parameter_name> = <value>, ...
```

Like the default value for parameters, <value> can be a numeric constant, a string enclosed in quotation marks, a time value such as 10 msec, target-language code enclosed in  $\{= \ldots =\}$ , or any of the list forms described in  $\underline{\text{Expressions}}$ .

#### Connections

Connections between ports are specified with the syntax:

```
<source_port_reference> -> <destination_port_reference>
```

where the port references are either <instance\_name>.<port\_name> or just <port\_name>, where the latter form is used for connections that cross hierarchical boundaries, as illustrated in the next section.

On the left and right of a connection statement, you can put a comma-separated list. For example, the above pair of connections can be written,

```
c.y, s.y \rightarrow s.x, t.x
```

A constraint is that the total number of channels on the left match the total number on the right. In addition, some targets require the types of all the ports to be the same.

A destination port (on the right) can only be connected to a single source port (on the left). However, a source port may be connected to multiple destinations, as in the following example:

```
reactor A {
  output y:int
}
reactor B {
  input x:int
}
main reactor {
  a = new A()
  b1 = new B()
  b2 = new B()
  a.y -> b1.x
  a.y -> b2.x
}
```



Lingua Franca provides a convenient shortcut for such multicast connections, where the above two lines can be replaced by one as follows:

$$(a.y) + -> b1.x, b2.x$$

The enclosing ( . . . )+ means to repeat the enclosed comma-separated list of sources however many times is needed to provide inputs to all the sinks on the right of the connection - > .

#### Import Statement

An import statement has the form:

```
import <classname> as <alias> from "<path>"
```

where <classname> and <alias> can be a comma-separated list to import multiple reactors from the same file. The <path> specifies another .lf file relative to the location of the current file. The as <alias> portion is optional and specifies alternative class names to use in the new statements.

# Hierarchy

Reactors can be composed in arbitrarily deep hierarchies. For example, the following program combines the Count and Scale reactors within on Container:

```
target Rust
import Count from "Count.lf"
import Scale from "Scale.lf"
import TestCount from "TestCount.lf"
reactor Container(stride: u32 = 2) {
  output y: u32
  c = new Count()
  s = new Scale(factor=stride)
  c.y -> s.x
  s.y -> y
}
main reactor Hierarchy {
  c = new Container(stride=4)
  t = new TestCount(stride=4, num_inputs=11)
  c.y -> t.x
}
```

▶ Lingua Franca diagram

The Container has a parameter named stride, whose value is passed to the factor parameter of the Scale reactor. The line

```
s.y \rightarrow y;
```

establishes a connection across levels of the hierarchy. This propagates the output of a contained reactor to the output of the container. A similar notation may be used to propagate the input of a container to the input of a contained reactor,

```
X \rightarrow S.X;
```

## **Connections with Logical Delays**

Connections may include a **logical delay** using the **after** keyword, as follows:

```
<source_port_reference> -> <destination_port_reference> after <time_value
where <time_value> can be any of the forms described in <a href="Expressions">Expressions</a>.
```

The **after** keyword specifies that the logical time of the event delivered to the destination port will be larger than the logical time of the reaction that wrote to source port. The time value is required to be non-negative, but it can be zero, in which case the input event at the receiving end will be one <u>microstep</u> later than the event that triggered it.

# **Physical Connections**

A subtle and rarely used variant of the -> connection is a **physical connection**, denoted ~> . For example:

```
main reactor {
  a = new A();
  b = new B();
  a.y ~> b.x;
}
```

This is rendered in by the diagram synthesizer as follows:



In such a connection, the logical time at the recipient is derived from the local physical clock rather than being equal to the logical time at the sender. The physical time will always exceed the logical time of the sender (unless fast is set to true), so this type of connection incurs a nondeterministic positive logical time delay. Physical connections are useful sometimes in <a href="Distributed-Execution">Distributed-Execution</a> in situations where the nondeterministic logical delay is tolerable. Such connections are more efficient because timestamps need not be transmitted and messages do not need to flow through through a centralized coordinator (if a centralized coordinator is being used).

#### **Reactions and Methods**

#### **Reaction Order**

A reactor may have multiple reactions, and more than one reaction may be enabled at any given tag. In Lingua Franca semantics, if two or more reactions of the same reactor are **simultaneously enabled**, then they will be invoked sequentially in the order in which they are declared. More strongly, the reactions of a reactor are **mutually exclusive** and are invoked in tag order primarily and declaration order secondarily. Consider the following example:

```
target Rust {
  timeout: 3 secs
}
main reactor Alignment {
  state s: u32 = 0
  timer t1(100 msec, 100 msec)
  timer t2(200 msec, 200 msec)
  timer t4(400 msec, 400 msec)
  reaction(t1) {=
    self.s += 1;
  =}
  reaction(t2) {=
    self.s -= 2;
  =}
  reaction(t4) {=
    println!("s = {}", self.s);
  =}
}
```

Every 100 ms, this increments the state variable s by 1, every 200 ms, it decrements s by 2, and every 400 ms, it prints the value of s. When these reactions align, they are invoked in declaration order, and, as a result, the printed value of s is always 0.

# **Overwriting Outputs**

Just as the reactions of the Alignment reactor overwrite the state variable s, logically simultaneous reactions can overwrite outputs. Consider the following example:

```
target Rust
reactor Overwriting {
  output y: u32
  state s: u32 = 0
  timer t1(100 msec, 100 msec)
  timer t2(200 msec, 200 msec)
  reaction(t1) -> y {=
    self.s += 1;
    ctx.set(y, self.s);
  =}
  reaction(t2) -> y {=
    self.s -= 2;
    ctx.set(y, self.s);
  =}
}
```

Here, the reaction to t1 will set the output to 1 or 2, but every time it sets it to 2, the second reaction (to t2) will overwrite the output with the value 0. As a consequence, the outputs will be 1, 0, 1, 0, ... deterministically.

#### Reacting to Outputs of Contained Reactors

A reaction may be triggered by the an input to the reactor, but also by an output of a contained reactor, as illustrated in the following example:

```
target Rust
import Overwriting from "Overwriting.lf"
main reactor {
    s = new Overwriting()
    reaction(s.y) {=
        let value = ctx.get(s__y).unwrap();
        if value != 0 && value != 1 {
            eprintln!("Output schould only be 0 or 1!");
            ctx.request_stop(Asap);
        }
        =}
}
```



This instantiates the above Overwriting reactor and monitors its outputs.

# **Triggering Contained Reactors**

A reaction can set the input of a contained reactor, thereby triggering its reactions, as illustrated in the following example:

```
target Rust
reactor Inside {
  input x: u32
  reaction(x) {=
    println!("Received {}", ctx.get(x).unwrap());
  =}
}
main reactor {
  i = new Inside()
  reaction(startup) -> i.x {=
    ctx.set(i_x, 42);
  =}
}
```

The reaction to **startup** declares the input port of the inside reactor as an effect and then sets it with value 42. This will cause the inside reactor's reaction to execute and print Received 42.

# Startup, Shutdown, and Reset Reactions

Reactions may be triggered by the special events startup, shutdown, or reset. For example,

```
reaction(startup) {=
  // ... Do something
=}
```

A reaction to **startup** is triggered at the very first tag when the program begins (or, if within a mode of a <u>modal reactor</u>, when the mode is first entered). This reaction will be logically simultaneous with reactions to <u>timers</u> that have a zero offset. As usual, for logically simultaneous reactions declared within the same reactor, the order in which they are invoked will be governed by the order in which they are declared.

A reaction to **shutdown** is invoked at program termination. See the <u>Termination</u> section for details.

Reactions to the **reset** event are not supported in Rust because <u>modal reactors</u> are not supported.

#### **Method Declaration**

The Rust target does not currently support methods.

# **Causality Loops**

# Cycles

The interconnection pattern for a collection of reactors can form a cycle, but some care is required. Consider the following example:

```
target Rust;
reactor A {
  input x:u32;
  output y:u32;
  reaction(x) -> y {=
    // ... something here ...
  =}
}
reactor B {
  input x:u32;
  output y:u32;
  reaction(x) {=
    // ... something here ...
  =}
  reaction(startup) -> y {=
    // ... something here ...
  =}
main reactor {
  a = new A();
  b = new B();
  a.y -> b.x;
  b.y \rightarrow a.x;
}
```

This program yields the following diagram:



The diagram highlights a **causality loop** in the program. At each tag, in reactor B, the first reaction has to execute before the second if it is enabled, a precedence indicated with the red dashed arrow. But the first can't execute until the reaction of A has executed, and that reaction cannot execute until the second reaction B has executed. There is no way to satisfy these requirements, so the tools refuse to generated code.

# Cycles with Delays

One way to break the causality loop and get an executable program is to introduce a <u>logical</u> <u>delay</u> into the loop, as shown below:

```
target Rust;
reactor A {
  input x:u32;
  output y:u32;
  reaction(x) -> y {=
   // ... something here ...
  =}
}
reactor B {
  input x:u32;
  output y:u32;
  reaction(x) {=
   // ... something here ...
  =}
  reaction(startup) -> y {=
   // ... something here ...
  =}
}
main reactor {
 a = new A();
 b = new B();
  a.y -> b.x after 0;
 b.y -> a.x;
Lingua Franca diagram
```

Here, we have used a delay of 0, which results in a delay of one <u>microstep</u>. We could equally well have specified a positive time value.

#### **Reaction Order**

Frequently, a program will have such cycles, but you don't want a logical delay in the loop. To get a cycle without logical delays, the reactions need to be reordered, as shown below:

```
target Rust;
reactor A {
  input x:u32;
  output y:u32;
  reaction(x) -> y {=
    // ... something here ...
  =}
reactor B {
  input x:u32;
  output y:u32;
  reaction(startup) -> y {=
    // ... something here ...
  =}
  reaction(x) {=
    // ... something here ...
  =}
}
main reactor {
  a = new A();
  b = new B();
  a.y \rightarrow b.x;
  b.y \rightarrow a.x;
}
```

Lingua Franca diagram					

There is no longer any causality loop.

# **Extending Reactors**

# Extending a Base Reactor

Lingua Franca supports defining a reactor class as an extension (or subclass), as in the following example:

```
target Rust;
reactor A {
  input a:u32;
  output out:u32;
  reaction(a) -> out {=
    ctx.set(out, ctx.get(a).unwrap());
  =}
}
reactor B extends A {
  input b:u32;
  reaction(a, b) -> out {=
    ctx.set(out, ctx.get(a).unwrap() + ctx.get(b).unwrap());
  =}
}
```



Here, the base class A has a single output that it writes to in reaction to an input. The subclass inherits the input, the output, and the reaction of A, and adds its own input b and reaction. When an input event a arrives, both reactions will be invoked, but, once again, in a well-defined order. The reactions of the base class are invoked before those of the derived class. So in this case, B will overwrite the output produced by A.

One limitation is that a subclass cannot have ports, actions, or state variables with the same names as those in the base class. The names must be unique.

A subclass can extend more than one base class by just providing a comma-separated list of base classes. If reactions in multiple base classes are triggered at the same tag, they will be invoked in the same order that they appear in the comma-separated list.

#### **Actions**

#### **Action Declaration**

An action declaration has one of the following forms:

```
logical action <name>(<min_delay>, <min_spacing>, <policy>)
physical action <name>(<min_delay>, <min_spacing>, <policy>)
```

The min\_delay, min\_spacing, and policy are all optional. If only one argument is given in parentheses, then it is interpreted as an min\_delay, if two are given, then they are interpreted as min\_delay and min\_spacing. The min\_delay and min\_spacing are time values. The policy argument is a string that can be one of the following: "defer" (the default), "drop", or "replace". Note that the quotation marks are needed.

If the action is to carry a payload, then a type must be given as well:

```
logical action <name>(<min_delay>, <min_spacing>, <policy>):<type>
physical action <name>(<min_delay>, <min_spacing>, <policy>):<type>
```

#### **Logical Actions**

Timers are useful to trigger reactions once or periodically. Actions are used to trigger reactions more irregularly. An action, like an output or input port, can carry data, but unlike a port, an action is visible only within the reactor that defines it.

There are two kinds of actions, **logical** and **physical**. A **logical action** is used by a reactor to schedule a trigger at a fixed logical time interval d into the future. The time interval d, which is called a **delay**, is relative to the logical time t at which the scheduling occurs. If a reaction executes at logical time t and schedules an action a with delay d, then any reaction that is triggered by a will be invoked at logical time t + d. For example, the following reaction schedules something (printing the current elapsed logical time) 200 msec after an input x arrives:

```
target Rust;
reactor Schedule {
  input x:u32;
  logical action a;
  reaction(x) -> a {=
    ctx.schedule(a, after!(200 ms));
  =}
  reaction(a) {=
    printf("
        Action triggered at logical time {} nsec after start.",
        ctx.get_elapsed_logical_time().as_nanos(),
     );
  =}
}
```

Lingua Franca diagram

Here, the delay is specified in the call to schedule within the target language code. Notice that in the diagram, a logical action is shown as a triangle with an **L**. Logical actions are always scheduled within a reaction of the reactor that declares the action.

The time argument is required to be non-negative. If it is zero, then the action will be scheduled one **microstep** later. See <u>Superdense Time</u>.

#### **FIXME**

An action may have a data type, in which case, a variant of the schedule() function can be used to specify a **payload**, a data value that is carried from where the schedule() function is called to the reaction that is triggered by the action. See the <u>Target Language Details</u>.

## **Physical Actions**

A **physical action** is used to schedule reactions at logical times determined by the local physical clock. If a physical action with delay *d* is scheduled at *physical* time *T*, then the *logical time* 

assigned to the event is T + d. For example, the following reactor schedules the physical action p to trigger at a **logical time** equal to the **physical time** at which the input  $\times$  arrives:

```
target Rust;
reactor Physical {
  input x:u32;
 physical action a;
  reaction(x) -> a {=
   let phys action = a.clone();
   ctx.spawn_physical_thread(move |link| {
     link.schedule(&phys action, Asap).unwrap();
   });
 =}
  reaction(a) {=
    println!(
      "Action triggered at logical time {} nsec after start.",
      ctx.get elapsed logical time().as nanos(),
    );
 =}
}
```



If you drive this with a timer, using for example the following structure:



then running the program will yield an output something like this:

```
Action triggered at logical time 201491000 nsec after start. Action triggered at logical time 403685000 nsec after start. Action triggered at logical time 603669000 nsec after start. ...
```

Here, logical time is lagging physical time by a few milliseconds. Note that, unless the <u>fast option</u> is given, logical time t chases physical time t, so t < t. Hence, the event being scheduled in the reaction to input  $\mathbf{x}$  is assured of being in the future in logical time.

Whereas logical actions are required to be scheduled within a reaction of the reactor that declares the action, physical actions can be scheduled by code that is outside the Lingua Franca system. For example, some other thread or a callback function may call schedule(), passing it a physical action. For example:

```
target Rust
main reactor {
  state start time: Instant = {= Instant::now() =}
  physical action a(100 msec): u32
  reaction(startup) -> a {=
    let phys action = a.clone(); // clone to move it into other thread
    // Start a thread to schedule physical actions.
    ctx.spawn physical thread(move |link| {
      loop {
        std::thread::sleep(Duration::from millis(200));
        link.schedule physical(&phys action, Asap).unwrap();
      }
    });
 =}
  reaction(a) {=
    let elapsed time = self.start time.elapsed();
    println!("Action triggered at logical time {} nsecs after start.", elap
  =}
}
```

Lingua Franca diagram

Physical actions are the mechanism for obtaining input from the outside world. Because they are assigned a logical time derived from the physical clock, their logical time can be interpreted as a measure of the time at which some external event occurred.

## **Triggering Time for Actions**

An action will trigger at a logical time that depends on the arguments given to the schedule function, the <min\_delay>, <min\_spacing>, and <policy> arguments in the action declaration, and whether the action is physical or logical.

For a **logical** action a, the tag assigned to the event resulting from a call to schedule() is computed as follows. First, let t be the current logical time. For a logical action, t is just the logical time at which the reaction calling schedule() is called. The **preliminary time** of the action is then just  $t + \min_{d \in \mathbb{N}} + \operatorname{coffset} > 0$ . This preliminary time may be further modified, as explained below.

For a **physical** action, the preliminary time is similar, except that t is replaced by the current *physical* time T when schedule() is called.

If a <min\_spacing> has been declared, then it gives a minimum logical time interval between the tags of two subsequently scheduled events. If the preliminary time is closer than <min\_spacing> to the time of the previously scheduled event (if there is one), then <policy> (if supported by the target) determines how the minimum spacing constraint is enforced.

The <policy> argument is currently not supported.

## Testing an Action for Presence

When a reaction is triggered by more than one action or by an action and an input, it may be necessary to test within the reaction whether the action is present. Just like for inputs, this can be done in the Rust target with ctx.is present(a), where a is the name of the action.

## **Superdense Time**

### Tag vs. Time

The model of time in Lingua Franca is a bit more sophisticated than we have hinted at. Specifically, a **superdense** model of time is used. In particular, instead of a **timestamp**, LF uses a **tag**, which consists of a **logical time** *t* and a **microstep** *m*.

A <u>logical action</u> may have a <min\_delay> of zero, and the <offset> argument to the schedule() function may be zero. In this case, the call to schedule() appears to be requesting that the action trigger at the *current logical time*. Here is where superdense time comes in. The action will indeed trigger at the current logical time, but one microstep later. Consider the following example:

```
target Rust
main reactor {
  state count: u32 = 1
  logical action a
  reaction(startup, a) -> a {=
    let tag = ctx.get tag();
    println!(
        "{}. Logical time is {}. Microstep is {}.",
        self.count,
        tag.offset from t0.as nanos(),
        tag.microstep(),
    );
    if self.count < 5 {</pre>
      self.count += 1;
      ctx.schedule(a, Asap);
    }
  =}
```

Lingua Franca diagram Executing this program will yield something like this:

```
    Logical time is 1649607749415269000. Microstep is 0.
    Logical time is 1649607749415269000. Microstep is 1.
    Logical time is 1649607749415269000. Microstep is 2.
    Logical time is 1649607749415269000. Microstep is 3.
    Logical time is 1649607749415269000. Microstep is 4.
```

Notice that the logical time is not advancing, but the microstep is (the logical time, in this case, gives the number of nanoseconds that have elapsed since January 1, 1970). The general rule is that **every** call to schedule() advances the tag by at least one microstep.

### **Logical Simultaneity**

Two events are **logically simultaneous** only if *both* the logical time and the microstep are equal. The following example illustrates this:

```
target Rust
reactor Destination {
  input x: u32
  input y: u32
  reaction(x, y) {=
    let tag = ctx.get tag();
    println!(
        "Time since start: {}, microstep: {}",
        tag.offset from t0.as nanos(),
        tag.microstep,
    );
    if ctx.is_present(x) {
      println!(" x is present.");
    if ctx.is_present(y) {
      println!(" y is present.");
   }
  =}
main reactor {
  logical action repeat
  d = new Destination()
  reaction(startup) -> d.x, repeat {=
    ctx.set(d x, 1);
    ctx.schedule(repeat, Asap);
  reaction(repeat) -> d.y {= ctx.set(d_y, 1); =}
}
```



The Destination reactor has two inputs, x and y, and it reports in a reaction to either input what is the logical time, the microstep, and which input is present. The main reactor reacts to startup by sending data to the x input of Destination. It then schedules a repeat action with an <offset> of zero. The repeat reaction is invoked strictly later, one microstep later. The output printed, therefore, will look like this:

```
Time since start: 0, microstep: 0
  x is present.
Time since start: 0, microstep: 1
  y is present.
```

The reported elapsed logical time has not advanced in the second reaction, but the fact that x is not present in the second reaction proves that the first reaction and the second are not logically simultaneous. The second occurs one microstep later.

## Alignment of Logical and Physical Times

Recall that in Lingua Franca, logical time "chases" physical time, invoking reactions at a physical time close to their logical time. For that purpose, the microstep is ignored.

## **Modal Reactors**

Modal Reactors are currently not supported in Rust.

### **Deadlines**

Lingua Franca includes a notion of a **deadline**, which is a constraint on the relation between logical time and physical time. Specifically, a program may specify that the invocation of a reaction must occur within some *physical* time interval of the *logical* time of the message. If a reaction is invoked at logical time 12 noon, for example, and the reaction has a deadline of one hour, then the reaction is required to be invoked before the physical-time clock of the execution platform reaches 1 PM. If the deadline is violated, then the specified deadline handler is invoked instead of the reaction.

### Purposes for Deadlines

A deadline in an LF program serves two purposes. First, it can guide scheduling in that a scheduler may prioritize reactions with deadlines over those without or those with longer deadlines. For this purpose, if a reaction has a deadline, then all upstream reactions on which it depends (without logical delay) inherit its deadline. Hence, those upstream reactions will also be given higher priority.

Second, the deadline mechanism provides a **fault handler**, a section of code to invoke when the deadline requirement is violated. Because invocation of the fault handler depends on factors beyond the control of the LF program, an LF program with deadlines becomes **nondeterministic**. The behavior of the program depends on the exact timing of the execution.

There remains the question of when the fault handler should be invoked. By default, deadlines in LF are **lazy**, meaning that the fault handler is invoked at the logical time of the event triggering the reaction whose deadline is missed. Specifically, the possible violation of a deadline is not checked until the reaction with the deadline is ready to execute. Only then is the determination made whether to invoke the regular reaction or the fault handler.

## Deadline Specification

A deadline is specified as follows:

WARNING: No source file found: ../code/rs/src/Deadline.lf

This reactor specifies a deadline of 10 milliseconds (this can be a parameter of the reactor). If the reaction to x is triggered later in physical time than 10 msec past the timestamp of x, then the second body of code is executed instead of the first. That second body of code has access to anything the first body of code has access to, including the input x and the output x and th

can be used to notify the rest of the system that a deadline violation occurred. This reactor can be tested as follows:

WARNING: No source file found: ../code/rs/src/DeadlineTest.lf



Running this program will result in the following output:

Normal reaction.

Deadline violation detected.

Deadline reactor produced an output.

The first reaction of the <code>Deadline</code> reactor does not violate the deadline, but the second does. Notice that the sleep in the <code>main</code> reactor occurs after setting the output, but because of the deterministic semantics of LF, this does not matter. The actual value of an output cannot be known until every reaction that sets that output <code>completes</code> its execution. Since this reaction takes at least 20 msec to complete, the deadline is assured of being violated.

Notice that the deadline is annotated in the diagram with a small clock symbol.

## Deadline Violations During Execution

Whether a deadline violation occurs is checked only *before* invoking the reaction with a deadline. What if the reaction itself runs for long enough that the deadline gets violated *during* the reaction execution? For this purpose, a target-language function is provided to check whether a deadline is violated during execution of a reaction with a deadline.

**NOTE**: As of this writing, this function is only implemented in the C target.

Consider this example:

WARNING: No source file found: ../code/rs/src/CheckDeadline.lf

## **Multiports and Banks**

Lingua Franca provides a compact syntax for ports that can send or receive over multiple channels and another syntax for multiple instances of a reactor class. These are respectively called **multiports** and **banks of reactors**.

## Multiports

To declare an input or output port to be a **multiport**, use the following syntax:

```
input[<width>] <name>:<type>;
output[<width>] <name>:<type>;
```

where <width> is a positive integer. This can be given either as an integer literal or a parameter name. Consider the following example:

```
target Rust;
reactor Source {
  output[4] out:usize;
  reaction(startup) -> out {=
    for (i, o) in out.into iter().enumerate() {
      ctx.set(o, i);
    }
  =}
reactor Destination {
  input[4] inp:usize;
  reaction(inp) {=
    let mut sum = 0;
    for i in inp {
      if let Some(v) = ctx.get(&i) {
        sum += v;
      }
    }
    println!("Sum of received: {}.", sum);
  =}
main reactor {
  a = new Source();
  b = new Destination();
  a.out -> b.inp;
}
Lingua Franca diagram
```

Executing this program will yield:

```
Sum of received: 6.
```

The Source reactor has a four-way multiport output and the Destination reactor has a four-way multiport input. These channels are connected all at once on one line, the second line from the last. Notice that the generated diagram shows multiports with hollow triangles. Whether it shows the widths is controlled by an option in the diagram generator.

The Source reactor specifies out as an effect of its reaction using the syntax -> out. This brings into scope of the reaction body a way to access the width of the port and a way to write to each channel of the port.

**NOTE**: In Destination, the reaction is triggered by in, not by some individual channel of the multiport input. Hence, it is important when using multiport inputs to test for presence of the input on each channel, as done above with the syntax:

```
if let Some(v) = ctx.get(&i) ...
```

An event on any one of the channels is sufficient to trigger the reaction.

#### Parameterized Widths

The width of a port may be given by a parameter. For example, the above Source reactor can be rewritten

```
reactor Source(width:int = 4) {
  output[width] out:int;
  reaction(startup) -> out {=
    ...
  =}
}
```

## Connecting Reactors with Different Widths

Assume that the Source and Destination reactors above both use a parameter width to specify the width of their ports. Then the following connection is valid:

```
main reactor {
   a1 = new Source(width = 3);
   a2 = new Source(width = 2);
   b = new Destination(width = 5);
   a1.out, a2.out -> b.in;
}
```

The first three ports of b will received input from a1, and the last two ports will receive input from a2. Parallel composition can appear on either side of a connection. For example:

```
al.out, a2.out -> b1.out, b2.out, b3.out;
```

If the total width on the left does not match the total width on the right, then a warning is issued. If the left side is wider than the right, then output data will be discarded. If the right side is wider than the left, then input channels will be absent.

Any given port can appear only once on the right side of the -> connection operator, so all connections to a multiport destination must be made in one single connection statement.

#### Banks of Reactors

Using a similar notation, it is possible to create a bank of reactors. For example, we can create a bank of four instances of Source and four instances of Destination and connect them as follows:

```
main reactor {
   a = new[4] Source();
   b = new[4] Destination();
   a.out -> b.in;
}
```



If the Source and Destination reactors have multiport inputs and outputs, as in the examples above, then a warning will be issued if the total width on the left does not match the total width on the right. For example, the following is balanced:

```
main reactor {
   a = new[3] Source(width = 4);
   b = new[4] Destination(width = 3);
   a.out -> b.in;
}
```

There will be three instances of Source, each with an output of width four, and four instances of Destination, each with an input of width 3, for a total of 12 connections.

To distinguish the instances in a bank of reactors, the reactor can define a parameter called **bank\_index** with any type that can be assigned a non-negative integer value (for example, int,  $size_t$ , or  $uint32_t$ ). If such a parameter is defined for the reactor, then when the reactor is instantiated in a bank, each instance will be assigned a number between 0 and n-1, where n is the number of reactor instances in the bank. For example, the following source reactor increments the output it produces by the value of bank index on each reaction to the timer:

```
target Rust
reactor MultiportSource(bank index: u32 = 0) {
  state bank index = bank index
  timer t(0, 200 \text{ msec})
  output out: u32
  state s: u32 = 0
  reaction(t) -> out {=
    ctx.set(out, self.s);
    self.s += self.bank index;
  =}
}
The width of a bank may also be given by a parameter, as in
main reactor(
  source bank width:int = 3,
  destination bank width:int = 4
) {
  a = new[source bank width] Source(width = 4);
  b = new[destination bank width] Destination(width = 3);
  a.out -> b.in;
}
```

#### **Contained Banks**

Banks of reactors can be nested. For example, note the following program:

```
target Rust;
reactor Child (
  bank index:usize = 0
) {
  state bank index = bank index;
  reaction(startup) {=
    println!("My bank index: {}.", self.bank index);
  =}
reactor Parent (
  bank index:usize = 0
) {
  c = new[2] Child();
}
main reactor {
  p = new[2] Parent();
}
```

Lingua Franca diagram

In this program, the Parent reactor contains a bank of Child reactor instances with a width of 2. In the main reactor, a bank of Parent reactors is instantiated with a width of 2, therefore, creating 4 Child instances in the program in total. The output of this program will be:

```
My bank index: 0.
My bank index: 1.
My bank index: 0.
My bank index: 1.
```

The order of these outputs will be nondeterministic if the execution is multithreaded (which it will be by default) because there is no dependence between the reactions, and, hence, they can execute in parallel.

The bank index of a container (parent) reactor can be passed down to contained (child) reactors. For example, note the following program:

```
target Rust
reactor Child(bank index: usize = 0, parent bank index: usize = 0) {
  state bank index = bank index
  state parent bank index = parent bank index
  reaction(startup) {=
    println!(
        "My bank index: {}. My parent's bank index: {}.",
        self.bank index,
        self.parent bank index,
    );
  =}
}
reactor Parent(bank index: usize = 0) {
  c = new[2] Child(parent bank index=bank index)
}
main reactor {
 p = new[2] Parent()
}
```

In this example, the bank index of the Parent reactor is passed to the parent\_bank\_index parameter of the Child reactor instances. The output from this program will be:

```
My bank index: 1. My parent's bank index: 1. My bank index: 0. My parent's bank index: 0. My bank index: 0. My parent's bank index: 1. My bank index: 1. My parent's bank index: 0.
```

Again, note that the order of these outputs is nondeterministic.

Finally, members of contained banks of reactors can be individually addressed in the body of reactions of the parent reactor if their input/output port appears in the reaction signature. For example, note the following program:

WARNING: No source file found: ../code/rs/src/ChildParentBank2.lf



Running this program will give something like the following:

```
Received 0 from child 0.
Received 1 from child 1.
Received 2 from child 0.
Received 3 from child 1.
```

FIXME: How to get the width of the bank in target code?

## Combining Banks and Multiports

Banks of reactors may be combined with multiports, as in the following example:

```
target Rust
reactor Source {
  output[3] out: usize
  reaction(startup) -> out {=
    for (i, o) in out.into iter().enumerate() {
      ctx.set(o, i);
    }
  =}
reactor Destination(bank index: usize = 0) {
  state bank_index = bank index
  input inp: usize
  reaction(inp) {=
    println!(
        "Destination {} received {}.",
        self.bank index,
        ctx.get(inp).unwrap(),
    );
  =}
}
main reactor MultiportToBank {
  a = new Source()
  b = new[3] Destination()
  a.out -> b.inp
}
Lingua Franca diagram
```

Lingua Franca diagram

The three outputs from the Source instance a will be sent, respectively, to each of three instances of Destination, b[0], b[1], and b[2]. The result of the program will be

something like the following:

```
Destination 0 received 0.
Destination 1 received 1.
Destination 2 received 2.
```

Again, the order is nondeterministic in a multithreaded context.

The reactors in a bank may themselves have multiports. In all cases, the number of ports on the left of a connection must match the number on the right, unless the ones on the left are iterated, as explained next.

#### **Broadcast Connections**

Occasionally, you will want to have fewer ports on the left of a connection and have their outputs used repeatedly to broadcast to the ports on the right. In the following example, the outputs from an ordinary port are broadcast to the inputs of all instances of a bank of reactors:

```
reactor Source {
  output out:int;
  reaction(startup) -> out {=
    ... write to out ...
 =}
reactor Destination {
  input in:int;
  reaction(in) {=
    ... read from in ...
  =}
}
main reactor ThreadedThreaded(width:int(4)) {
  a = new Source();
  d = new[width] Destination();
  (a.out) + -> d.in;
}
```

The syntax (a.out)+ means "repeat the output port a.out one or more times as needed to supply all the input ports of d.in." The content inside the parentheses can be a commaseparated list of ports, the ports inside can be ordinary ports or multiports, and the reactors inside can be ordinary reactors or banks of reactors. In all cases, the number of ports inside the parentheses on the left must divide the number of ports on the right.

## **Interleaved Connections**

The **interleaved** keyword is not supported by Rust.

## **Generic Reactors**

Generic reactors are not supported in Rust.

## **Preambles**

## Preamble

Reactions may contain arbitrary target-language code, but often it is convenient for that code to invoke external libraries or to share procedure definitions. For either purpose, a reactor may include a **preamble** section.

FIXME: Add preamble example.

# **Distributed Execution**

Federated execution is not supported in Rust.

### **Termination**

#### **Shutdown Reactions**

There are several mechanisms for terminating a Lingua Franca in an orderly fashion. All of these mechanisms result in a **final tag** at which any reaction that declares **shutdown** as a trigger will be invoked (recall that a **tag** is a tuple (**logical time**, **microstep**)). Other reactions may also be invoked at this final tag, and the order in which reactions are invoked will be constrained by the normal precedence rules.

If a reaction triggered by **shutdown** produces outputs, then downstream reactors will also be invoked at the final tag. If the reaction schedules any actions by calling **schedule()**, those will be ignored. In fact, any event after the final tag will be ignored. After the completion of the final tag, the program will exit.

There are four ways to terminate a program:

- **Timeout**: The program specifies the last logical time at which reactions should be triggered.
- **Starvation**: At the conclusion of some tag, there are no events in the event queue at future tags.
- **Stop request**: Some reaction requests that the program terminate.
- External signal: Program is terminated externally using operating services like control-C or kill.

We address each of these in turn.

#### **Timeout**

The <u>target property timeout</u> specifies the last logical time at which reactions should be triggered. The last invocation of reactions will be at tag (timeout, 0).

There is a significant subtlety when using <u>physical connections</u>, which are connections using the syntax ~> . Such connections specify that the tag at the receiving end will be based on the physical time at which the message is received. If the tag assigned at the receiving end is greater than the final tag, then the message is lost. Hence, **messages sent near the timeout time are likely to be lost!** 

#### Starvation

If a Lingua Franca program has no <u>physical actions</u>, and if at any time during execution there are no future events waiting to be processed, then there is no possibility for any more reactions to occur and the program will exit. This situation is called **starvation**. If there is a **timer** anywhere in the program with a period, then this condition never occurs.

One subtlety is that reactions triggered by **shutdown** will be invoked one microstep later than the last tag at which there was an event. They cannot be invoked at the same tag because it is only after that last tag has completed that the runtime system can be sure that there are no future events. It would not be correct to trigger the **shutdown** reactions at that point because it would be impossible to respect the required reaction ordering.

### Stop Request

If a reaction calls the built-in <code>request\_stop()</code> function, then it is requesting that the program cease execution as soon as possible. This cessation will normally occur in the next microstep. The current tag will be completed as normal. Then the tag will be advanced by one microstep, and reactions triggered by <code>shutdown</code> will be executed, along with any other reactions with triggers at that tag, with all reactions executed in precedence order.

## **External Signal**

A control-C or other kill signal to a running Lingua Franca program will cause execution to stop immediately.